

## MATH 308 Sheet 4

### Higher Order Equations

Second and higher order equations are more complicated than first order equations, but they may be solved directly or approximated numerically with the same Maple tools that we used for first order problems.

**Example 1.** Solve the initial value problem

$$y'' + 2y' + y = x^2 + 1 + e^x, \quad y(0) = 0, \quad y'(0) = 2.$$

First clear Maple's memory

```
> restart;
```

Assign the differential equation the name `de1` (Notice how  $y''$  is entered as `diff(y(t),t$2)`).

```
> de1:=diff(y(x),x$2)+2*diff(y(x),x)+y(x)=x^2+1+exp(x);
```

```
> inits1:=y(0)=0,D(y)(0)=2;
```

To get the general solution to the differential equations, we use

```
> sol1:=dsolve(de1,y(x));
```

To solve the initial value problem, we use

```
> dsolve({de1,inits1},y(x));
```

If you prefer a different from the answer, try

```
> expand(%);sol1s:=simplify(%exp);
```

To plot the solution of the initial value problem on the interval  $[-3, 3]$ , use

```
> plot(rhs(sol1s),x=-3..3);
```

To obtain a floating point decimal value for the solution at  $x = 2.3$ , use

```
> subs(x=2.3,sol1s);
```

```
> evalf(%);
```

**Example 2.** Solve the nonlinear initial value problem

$$t^2 y'' - yy' = y^{-3}, \quad y(1) = 0.5, \quad y'(1) = 2.1.$$

```
> restart;
```

```
> de2:=t^2*diff(y(t),t$2)-y(t)*diff(y(t),t)=y(t)^(-3);
```

```
> inits2:=y(1)=0.5,D(y)(1)=2.1;
```

```
> sol2:=dsolve({de2,inits2},y(t),numeric);
```

The approximate values of the solution and of the derivative of the solution at  $x = 1.3$  are given by

```
> sol2(1.3);
```

The easiest way to plot a numerical solution on the interval  $[0.1, 1.5]$  is to use `plots[odeplot]`.

```
> with(plots):
```

```
> odeplot(sol2,[t,y(t)],0.1..1.5);
```

## Homogeneous equations

**Example 3.** Find the general solution of

$$y'' + y' + y = 0.$$

```
> restart:
```

```
> de3:=diff(y(x),x$2)+diff(y(x),x)+y(x)=0;
```

```
> sol3:=dsolve(%,y(x));
```

The individual functions which comprise the fundamental set of solutions to this equation can be obtained by

```
> y1:=rhs(subs(_C2=0,_C1=1,sol3));
```

```
> y2:=rhs(subs(_C2=1,_C1=0,sol3));
```

**Remark.** For higher than 3 order equations Maple's `dsolve` is not adequate.

## Undetermined Coefficients

**Example 4.** Let us use Maple to help find a particular solution for  $y'' + 4y = t \cos(2t)$ . There are only two new commands. The command `solve` tries to solve equations exactly in contrast to `fsolve` which is an approximation scheme. The other command is `identity`, which is used to equate coefficients of like functions of  $t$ . Our guess for the particular solution is

$$y_p = t(At + B) \cos(2t) + t(Ct + D) \sin(2t).$$

This would be very painful to do by hand!

**WARNING:** Don't use  $I$  as an undetermined coefficient—Maple understands this to be  $\sqrt{-1}$ .

```
> restart:
```

Enter the differential equation

```
> de4:=diff(y(t),t$2)+4*y(t)=t*cos(2*t);
```

Solving the homogeneous equation is easy.

```
> hsol4:=dsolve(lhs(de4),y(t));
```

Now enter your guess for the particular solution  $y_p$

```
> yp:=t*(A*t+B)*cos(2*t)+t*(C*t+D)*sin(2*t);
```

Plug it into the differential equation and expand

```
> subs(y(t)=yp,de4);  
> simplify(%);
```

Equate coefficients of like functions of  $t$

```
> identity(%,t);
```

The output looks funny, but we can now solve for the coefficients and assign the list to the variable `undetcoeffs` (or whatever name you like)

```
> undetcoeffs:=solve(%,{A,B,C,D});
```

Plug them into  $y_p$

```
> yp:=subs(undetcoeffs,yp);
```

How nice!!

You should check the answer

```
> subs(y(t)=yp,de4);simplify(%);
```

## Variation of parameters

**Example 5.** Find the general solution to the equation

$$y'' + 4y = \sin^2(2t).$$

```
> restart;
```

We must first find the solutions to the homogeneous equation.

```
> de5:=diff(y(t),t$2)+4*y(t)=(sin(2*t))^2;  
> hsol5:=dsolve(lhs(de5),y(t));
```

Replace the constants in the solution to the homogeneous equation by functions

```
> psol5:=subs(_C1=u1(t),_C2=u2(t),hsol5);
```

The derivatives  $u'_1$  and  $u'_2$  satisfy two equations:

$$u'_1 y_1 + u'_2 y_2 = 0,$$

$$u'_1 y'_1 + u'_2 y'_2 = \sin^2(2t).$$

We denote  $u'_1$  and  $u'_2$  by  $u1p$  and  $u2p$  respectively; and we set up the equations.

```
> y1:=cos(2*t):y2:=sin(2*t):  
> eq1:=u1p*y1+u2p*y2=0;  
> eq1:=u1p*diff(y1,t)+u2p*diff(y2,t)=(sin(2*t))^2;
```

We then ask Maple to solve

```
> solve({eq1,eq2},{u1p,u2p}):  
> convert(% ,exp):  
> upsol:=simplify(%);
```

We need to integrate to find  $u_1(t)$  and  $u_2(t)$ .

```
> subs(upsol,u1p):  
> int(% ,t):  
> u1(t):=simplify(%);  
> subs(upsol,u2p):  
> int(% ,t):  
> u2(t):=simplify(%);
```

We can get the particular solution:

```
> y(t)=rhs(psol5):  
> psol:=simplify(%);
```

Adding the solution to the homogeneous equation, we get a general solution:

```
> rhs(psol)+rhs(hsol):  
> gsol:=y(t)=simplify(%);
```